

A Software Algorithm for Implementing Automatic Power Ranging

This product note presents a software algorithm for performing automatic power ranging in the HP 8960 Series 10 wireless communication test set with the HP E1960A GSM mobile test application installed. HP E1960A GSM mobile test application revisions, up to and including, A.01.04, must use this algorithm to implement automatic power ranging (autoranging).

When Is Automatic Power Ranging Required?

Automatic power ranging is required when the input signal's power level is much different than the power level expected by the test set. For example: an EGSM mobile station, when commanded to transmit at MS TX level 15, should have an transmit power level of 13 dBm. The power level expected by the test set is tied automatically to the value of the MS TX Level field. If the mobile station's transmit power level was in fact -20 dBm, then the test set's RF receiver would have to be manually configured to make measurements with specified accuracy at the input signal's true power level.


When the mobile station's transmit power level is known to be within ± 3 dB of its expected value, no autoranging is necessary. The linkage between the test set's MS TX Level field and the Expected Power field automatically configures the RF receiver to measure the input signal with specified measurement accuracy at the current expected power setting. Typically the mobile station's transmit power level is well behaved once it has been calibrated.

One of the first steps in the manufacturing process is the calibration of the mobile station's transmit power levels. During initial power calibration, it is very likely that the mobile station's transmit power level may be much different than the power level expected by the test set. Under these circumstances and,

with no autoranging in the test set, it is necessary to verify that the test set's RF receiver is properly configured to make measurements with specified accuracy at the input signal's power level. Measurements performed with the RF receiver not properly configured may not meet their specified measurement accuracy.

To eliminate the possibility of obtaining measurement results with the test set's RF receiver not properly configured, you should implement automatic power ranging using external software.

**HP 8960 Series 10
E5515A Wireless Communications Test Set**





Contents

When Is Automatic Power Ranging Required?	1
Who Should Read This Product Note?	2
RF Receiver Overview	3
Power Ranging Algorithm	5
Transmit Power Measurement Setup Configuration For Autoranging	7
Avoiding Timeouts	7
HP BASIC Program	8
The Program	9

Who Should Read This Product Note?

This product note is intended for manufacturing production engineers. It is assumed that the reader is familiar with the basic operation and programming of the test set. Further, it is assumed that the reader is

familiar with mobile station manufacturing. For demonstration purposes, the algorithm is implemented using the HP BASIC programming language, also known as RMB or Rocky Mountain BASIC.



RF Receiver Overview

The test set's RF receiver consists of a measurement downconverter and a demodulation downconverter as shown in figure 1. The RF receiver's gain and attenuation level settings are based upon:

1. The power level the test set expects the mobile station to be transmitting (the expected power).
2. A user supplied amplitude offset value applied to the RF IN/OUT connector. See the Amplitude Offset (RF IN/OUT port) parameter found on the System Config screen.

The level of expected power is set through the **Expected Power** field (on screen 3 of 3 in the Call Params column) found on the Call Setup or Measurement screens, see figure 2. The value of the **Expected Power** field is automatically linked to the mobile station's transmit power level setting (the **MS TX Level** field on screen 2 of 3), or it can be set by the user, either manually or programmatically. For any measurement, the power level of the input signal, relative to the expected power level, can produce one of four possible measurement conditions (see figure 3):


- An **overrange condition**, where the signal power seen by the analog-to-digital converter (ADC), as determined by the RF receiver's gain and attenuation level settings, is above the accuracy limits of the ADC. A measurement integrity value of 5 is returned when an overrange condition exists. If the measurement is successful, the measured result does not meet specified measurement accuracy. An overrange condition occurs when the input signal's power level is (nominally) 3 to 5 dB higher than the expected power level.
- A **measurable condition**, where the signal power seen by the ADC, as determined by the RF receiver's gain and attenuation level settings, lies within the accuracy limits of the ADC. A measurement integrity value of 0 is returned when a


measurable condition exists. If the measurement is successful the measured value may or may not meet specified measurement accuracy. If the measured result is within ± 3 dB of the expected power setting, the measured result meets specified measurement accuracy.

Note 1: The input signal's power level may be in the measurable range but the measurement may fail for other reasons, such as synchronization issues, signal related issues, hardware related issues, or no trigger. In these cases a measurement integrity value other than 0, 2, 5, or 6 will be returned.

Note 2: It is possible to make a successful measurement (that is, have an integrity value of 0 returned with the measured results) with the input signal's power level in the measurable range but not within ± 3 dB of the expected power setting. In this case the measured results are valid but do not meet specified measurement accuracy.

- An **underrange condition**, where the signal power seen by the ADC, as determined by the RF receiver's gain and attenuation level settings, is below the accuracy limits of the ADC. A measurement integrity value of 6 is returned when an underrange condition exists. If the measurement is successful, the measured result does not meet specified measurement accuracy. An underrange condition occurs when the input signal's power level is (nominally) 5 to 10 dB lower than the expected power level.
- A **timeout condition** (not shown in figure 3), caused by some characteristic of the input signal or an internal hardware problem, which prevents the measurement from completing. The measurement is not successful, the measured result is set to NAN (not a number = 9.91E+37), and an integrity value of 2 is returned along with the measurement result.





flowchart3.eps

Figure 4. Power Ranging Algorithm Flowchart

Power Ranging Algorithm

The power ranging algorithm presented in figure 4 can be used to autorange the test set. Users may decide to modify the technique to match their needs. The technique involves making a series of transmit power measurements to range the RF receiver to the power level of the input signal before making any other measurements for which specified measurement accuracy is desired. The transmit power measurement is a very fast measurement, with a typical measurement time of 11 milliseconds for a single burst measurement.

The measurement integrity indicator is used to determine which of the four signal measurement conditions exists after a measurement has been made. If the measured result has not been obtained under acceptable conditions the expected power is adjusted and another measurement is made. The specific steps in the algorithm are:

1. Get current transmit power measurement setup configuration and save for use upon exit.
2. Get current expected power level setting for use in calculating new expected power level settings.
3. Get the RF IN/OUT connector amplitude offset value and save for use in calculating expected power levels. Calculate the minimum and maximum allowable expected power level using the RF IN/OUT connector's amplitude offset value and state.
 - The minimum receiver power for the transmit power measurement is -20 dBm. Using the RF offset value ($\pm x.x$ dB) and the RF offset state (ON = 1, OFF = 0) calculate the minimum expected power level as follows:
$$\text{Min exp pwr} = -20 - (\text{Offset value} \times \text{Offset state})$$
 - The maximum receiver power for the transmit power measurement is +43 dBm. Using the RF offset value ($\pm x.x$ dB) and the RF offset state (ON = 1, OFF = 0) calculate the maximum expected power level as follows:
$$\text{Max exp pwr} = +43 - (\text{Offset value} \times \text{Offset state})$$
4. Configure transmit power measurement for autoranging.
5. Make a transmit power measurement and read the measurement integrity value and the measured result.

6. Check the measurement integrity value:
 - 6a. If Integrity = 0: Measurement completed with no error conditions. The measured value is checked to see if it is greater than or equal to the minimum expected power level calculated in step 3. This check is done to ensure that the input signal's power level can be measured with specified measurement accuracy at the minimum expected power level setting.
 - If the measured power is greater than or equal to the minimum expected power the following actions are taken:
 - * Expected power is set to maximum expected power *if* measured power is greater than the maximum expected power, *otherwise*, the expected power is set to measured power
This new expected power level should be used as the expected power for any subsequent measurements.
 - * Set transmit power measurement setup parameters back to settings obtained in step 1.
 - * Exit autorange algorithm
 - If the measured power is less than the minimum expected power the signal cannot be measured with specified measurement accuracy. The following actions are taken:
 - * Exit autorange algorithm with error condition.
 - 6b. If Integrity = 2: A measurement timeout has occurred. To recover, set the expected power to produce the RF receiver's minimum power level setting of -20 dBm and remeasure. Actions taken:
 - Set new expected power level to the minimum expected power level calculated in step 3.
 - Remeasure (go to step 5).
- Note:** The user may wish to keep track of how many times the measurement has timed out and terminate the algorithm when some predefined number has been reached. This will prevent the algorithm from trying indefinitely to autorange on an unmeasurable signal.

6c. If Integrity = 5: An overrange condition has occurred. The input signal power level is greater than 3 dB above the expected power level setting. The underrange threshold level is approximately 10 dB below the expected power level setting. Adjust the expected power level up by a delta of 13 dB (3 + 10) and remeasure. Actions taken:

- Calculate a new expected power level setting as follows:
$$\text{New exp pwr} = \text{Old exp pwr} + 13$$
- If New exp pwr is greater than the maximum expected power level calculated in step 3, set New exp pwr to maximum expected power level.
- Set the expected power to its new value
- Remeasure (go to step 5).

Note: The primary concern with an overrange condition is the absolute power level of the input signal. Damage to the test set will result if it is above the maximum input level for the RF IN/OUT port and it is not removed. The user should keep track of how many times the measurement has overranged and terminate the algorithm when some predefined number has been reached. This will prevent the algorithm from trying indefinitely to autorange on an unmeasurable signal.

6d. If Integrity = 6: An underrange condition has occurred. The input signal level is greater than 5 to 10 dB below the expected power level. Set the expected power to the power level just measured and remeasure. Actions taken:

- If measured power is less than the minimum expected power level calculated in step 3, set the new expected power to the minimum expected power level, otherwise set the new expected power to the measured power.

- Remeasure (go to step 5).

Note: The primary concern with an underrange condition is that the power level of the input signal may be too low to measure with specified measurement accuracy. This condition occurs when the measured value is below the minimum expected power level calculated in step 3. If the expected power is already set to its minimum value it is impossible to autorange to the input signal's power level. The user may wish to keep track of how many times the measurement has underranged and terminate the algorithm when some predefined number has been reached. This will prevent the algorithm from trying indefinitely to autorange on an unmeasurable signal.

Thus, the measurement integrity value is checked for any underrange, overrange, or timeout conditions caused by the power level of the input signal. The expected power is iteratively reset to match the input signal's power level until a normal value for measurement integrity is obtained. Once a normal integrity value is obtained the measured power level is checked to verify that measurements can be made with specified measurement accuracy. If measurements can be made an expected power level is calculated, set, and the algorithm terminates. If measurements cannot be made the algorithm terminates with an error condition.

An upper limit for the number of iterations made by the autoranging algorithm should be determined based upon the user's application. If it is normal for the mobile station's transmit power level to be far outside the nominal range set by radio test standards, one might decide to let the algorithm execute many iterations. Generally, if the signal cannot be ranged with 3-5 iterations, the input signal should be considered non-ideal, and further attempts to autorange using this algorithm should not be made.

After the test set's RF receiver has been configured for the input signal's power level using this algorithm, additional measurements can be made at the same input signal power level. Whenever the input signal's power level changes, the algorithm can be repeated before making new measurements to ensure that the RF receiver has been properly configured.

Transmit Power Measurement Setup Configuration For Autoranging

For the autoranging algorithm the transmit power measurement is set up as follows:

- Multi-Measurement Count: Off
- Trigger Arm: Single
- Trigger Source: Auto

Note: When Auto is selected as the trigger source for a measurement, the test set will automatically select the correct trigger source (Protocol or RF Rise) for the current operating mode (that is, Active Cell operating mode or Test mode operating mode). The default trigger source for Active Cell operating mode is Protocol. The default trigger source for Test mode operating mode is RF Rise.

- Trigger Delay: 0
- Trigger Qualification: Off

Note: Do not set to On. If trigger qualification is set to On AND measurement timeouts are off AND no signal is present, the measurement will wait forever.

- Measurement Timeout: greater than or equal to 1 second

Note: Do not set to Off. If timeouts are off AND a measurement never completes due to a signal or internal hardware problem, the measurement will wait forever.

Avoiding Timeouts

Measurement timeouts are costly in terms of measurement execution time. However, measurement timeouts are necessary in a general purpose autoranging algorithm to avoid having the algorithm hang forever trying to autorange a signal that cannot be measured. In situations where there is a strong possibility that the input signal power level may be well below the expected power level, as in the case of calibrating a mobile stations's transit power levels for the first time, the potential for encountering a timeout condition can be minimized by setting the expected power to its minimum level before executing the autoranging algorithm. This value can be calculated as shown in step 3.

HP BASIC Program

The example program on the following pages demonstrates one possible implementation of the autoranging algorithm. In this example, the algorithm is implemented as a function. When the function is called it returns a pass/fail flag which the calling context can use to make a measure/no measure decision. In order to maximize the execution speed of the algorithm every attempt is made to minimize HP-IB bus transactions and minimize the number of characters sent during any bus transaction. Therefore compound and complex commands are used extensively.

The following checks and features have been implemented inside the algorithm:

- If the input signal's power level cannot be measured with specified measurement accuracy the algorithm exits with a fail flag.
- Only two timeouts are allowed. The algorithm exits with a fail flag after the second timeout.
- If an overrange condition exists AND the expected power has already been set to the maximum expected power level setting the algorithm exits with a fail flag.
- If an underrange condition exists AND the expected power has already been set to the minimum expected power level setting the algorithm exits with a fail flag.
- An error message facility has been added to globally pass explanatory error messages back to the calling context.

The Program

```
10  ! This example is intended to show one possible implementation of the
20  ! autoranging algorithm. The autoranging algorithm is contained in the
30  ! function FNAuto_range.
40  !
50  ! The code necessary to configure the base station emulator and bring up
60  ! a call is not shown in this example. The user can add code to configure
70  ! the base station emulator and establish a call, or simply bring up a
80  ! call manually and then run this program.
90  !
100 OPTION BASE 1
110 COM /Address/ T_s ! Holds address of the test set.
120 COM /Messages/ Aut_rng_err_msg$[80] ! Holds autorange error messages.
130 REAL Txp_array(2),Pfer_array(4)
140 INTEGER Integrity,Traf_chan,Ms_pwr_lvl
150 T_s=714 ! Test set's HP-IB address.
160 CLEAR SCREEN
170 ! The following section of code steps through a low, medium and high channel
180 ! and tests transmit power and phase&frequency error at a low, medium and high
190 ! transmit power level at each channel.
200 !
210 ! Configure TX Power Measurement:
220 !
230 OUTPUT T_s;"set:txp:cont off;tim 5;trig:sour prot;qual on"
240 !
250 ! Configure Phase & Frequency Error Measurement:
260 !
270 OUTPUT T_s;"set:pfer:cont off;tim 5;bsyn mid;trig:sour prot;qual on"
280 !
290 FOR Traf_chan=1 TO 123 STEP 61 ! Test a low, medium and high channel
300   OUTPUT T_s;"call:tch:seq ";Traf_chan ! Set traffic channel.
310   OUTPUT T_s;"call:stat:stat?" ! Verify that the call is still in
320   ENTER T_s;Call_status$ ! the connected state after handover.
330   IF Call_status$<>"CONN" THEN
340     PRINT "Call handover failed. Program terminated."
350     STOP
360   END IF
370   FOR Ms_pwr_lvl=5 TO 15 STEP 5 ! Test power levels 5, 10 & 15.
380     OUTPUT T_s;"call:ms:txl:seq ";Ms_pwr_lvl
390     ! Auto range to input signal's power level before making measurements.
400     IF (FNAuto_range(Integrity)) THEN
410       OUTPUT T_s;"init:txp;pfer::fetc:txp?;pfer?"
420       ENTER T_s;Txp_array(*),Pfer_array(*)
430       PRINT Txp_array(*),Pfer_array(*)
440     ELSE
450       PRINT Aut_rng_err_msg$
460       PRINT "The measurement integrity value is ";Integrity
470       PRINT "TCH = ";Traf_chan;" MS TX Level = ";Ms_pwr_lvl
480     END IF
490   NEXT Ms_pwr_lvl
500 NEXT Traf_chan
510 !
520 END ! End of program
530 !
540 DEF FNAuto_range(INTEGER Integrity)
550   ! Local variable name definitions:
560   ! Count_num(Cn),Count_state(Cs),Trig_qual(Tq),Continuous(Ct),Timeout_state(Ts)
570   ! Rf_offset_state(Rs),Trig_del(Td),Timeout_time(Tt),Rf_offset(Ro),Expected_pwr(Ep)
580   ! Measured_pwr(Mp),Expected_pwr_low(Ep_l),Expected_pwr_high(Ep_h)
590   ! Dimension and initialize local variables:
600   COM /Address/ T_s
610   COM /Messages/ Aut_rng_err_msg$[80]
```

```

620 INTEGER Cn,Cs,Tq,Ct,Ts,Rs,Timed_out,High_power,Low_power
630 REAL Td,Tt,Ro,Ep,Mp,Ep_l,Ep_h
640 DIM Ts$[18] ! Trig_sour$(Ts$)
650 Timed_out=0 ! Measurement timeout flag.
660 High_power=0 ! Input power too high to measure flag.
670 Low_power=0 ! Input power too low to measure flag.
680 Aut_rng_err_msg$="" ! Clear out any previous error message.
690 !
700 ! Get current transmit power measurement setup conditions.
710 OUTPUT T_s;"set:txp:cont?;coun:numb?;stat?::set:txp:tim:time?;stat?
;:set:txp:trig:del?;qual?;sour?"

720 ENTER T_s;Ct,Cn,Cs,Tt,Ts,Td,Tq,Ts$
730 !
740 ! Get amplitude offset information and current expected power setting.
750 OUTPUT T_s;"syst:corr:gain?;stat?::rfan:exp:pow?"
760 ENTER T_s;Ro,Rs,Ep
770 Ep_l=-20-(Ro*Rs) ! Calculate minimum allowable setting of expected power
780 Ep_h=+43-(Ro*Rs) ! Calculate maximum allowable setting of expected power
790 !
800 ! Setup transmit power measurement for autoranging.
810 OUTPUT T_s;"set:txp:cont 0;tim 1;coun:stat off
;:set:txp:trig:sour auto;del 0;qual 0"

820 !
830 LOOP ! Auto range to input signal power level.
840 OUTPUT T_s;"read:txp?"
850 ENTER T_s;Integrity,Mp
860 SELECT Integrity
870 CASE 0 ! Normal measurement
880 IF (Mp>=Ep_l) THEN ! Input signal's power can be measured with specified
accuracy
890 IF (Mp>Ep_h) THEN
900 Ep=Ep_h
910 ELSE
920 Ep=Mp
930 END IF
940 OUTPUT T_s;"set:txp:cont ";Ct;" ;tim:time ";Tt;" ;stat ";Ts;"
;:set:txp:coun:numb ";Cn;" ;stat ";Cs
950 OUTPUT T_s;"set:txp:trig:sour ";Ts$;" ;del ";Td;" ;qual ";Tq;"
;:rfan:exp:pow ";Ep

960 Aut_rng_err_msg$="No error."
970 RETURN 1
980 ELSE ! Cannot set expected power to >=Ep_l.
990 Aut_rng_err_msg$="Signal cannot be measured with specified accuracy."
1000 RETURN 0
1010 END IF
1020 CASE 2 ! Measurement timed out.
1030 IF Timed_out THEN
1040 Aut_rng_err_msg$="Measurement timed out."
1050 RETURN 0
1060 ELSE
1070 Ep=Ep_l
1080 Timed_out=1
1090 END IF
1100 CASE 5 ! Input signal power overrange condition.
1110 IF High_power THEN
1120 Aut_rng_err_msg$="CAUTION: Input signal power too HIGH to measure!"
1130 RETURN 0
1140 ELSE
1150 Ep=Ep+13
1160 IF Ep>Ep_h THEN
1170 Ep=Ep_h
1180 High_power=1
1190 END IF

```

```

1200     END IF
1210 CASE 6 ! Input signal power underrange condition.
1220     IF Low_power THEN
1230         Aut_rng_err_msg$="Input signal power too low too measure."
1240         RETURN 0
1250     ELSE
1260         IF Mp<Ep_1 THEN
1270             Ep=Ep_1
1280             Low_power=1
1290         ELSE
1300             Ep=Mp
1310         END IF
1320     END IF
1330 CASE ELSE ! Some other measurement problem exists.
1340     Aut_rng_err_msg$="A measurement problem exists."
1350     RETURN 0
1360 END SELECT
1370 OUTPUT T_s;"rfan:exp:pow ";Ep
1380 END LOOP
1390 FNEND

```



Expanding Possibilities

For more information about the HP 8960 Series 10 E5515A Wireless Communications Test Set visit our web site at:

<http://www.hp.com/hp8960support>

Available literature includes: brochures, technical specifications, product notes, configuration guide, and case study.

Warranty This test set is backed by Hewlett-Packard with a one-year warranty and world class support services. With HP sales and service centers worldwide, you can be assured of getting the global support that you need.

For more information about Hewlett-Packard test and measurement products, applications, services, and a current sales office listing, visit our web site at:

<http://www.hp.com/go/tmdir>

You can also contact one of the following centers and ask for a test and measurement sales representative.

United States:

Hewlett-Packard Company
Test and Measurement Call Center
P.O. Box 4026
Englewood, CO 80155-4026
(tel) 1 800 452 4844

Canada:

Hewlett-Packard Canada Ltd.
5150 Spectrum Way
Mississauga, Ontario
L4W 5G1
(tel) 1 877 894 4414

Europe:

Hewlett-Packard Company
European Marketing Organisation
P.O. Box 999
1180 AZ Amstelveen
The Netherlands
(tel) (31 20) 547 9999

Japan:

Hewlett-Packard Japan Ltd.
Measurement Assistance Center
9-1, Takakura-Cho, Hachioji-Shi,
Tokyo 192-8510, Japan
(tel) (81) 426 56 7832
(fax) (81) 426 56 7840

Latin America:

Hewlett-Packard
Latin American Region Headquarters
5200 Blue Lagoon Drive, 9th Floor
Miami, Florida 33126 U.S.A.
(tel) (305) 267-4245
(tel) (305) 267-4220
(fax) (305) 267-4288

Australia/New Zealand:

Hewlett-Packard Australia Ltd.
31-41 Joseph Street
Blackburn, Victoria 3130
Australia
(tel) 1 800 629 485 (Australia)
(tel) 0800 738 378 (New Zealand)
(fax) (61 3) 9210 5489

Asia Pacific:

Hewlett-Packard Asia Pacific Ltd.
17-21/F Shell Tower, Times Square,
1 Matheson Street, Causeway Bay,
Hong Kong, SAR
(tel) (852) 2599 7777
(fax) (852) 2506 9285